

Assignment 6

For this assignment we will finalise the domain-specific language that we use to model the scrabble boards we will be playing on. More precisely we will add

1. arithmetic division
2. arithmetic modulo
3. explicit variable declarations
4. variable shadowing (if-statements and while-loops will allow local variable declarations with their sub-statements). This requirement is necessary in order to have sensible variable declarations as variables declared within a loop would otherwise persist outside of the loop.
5. Restricted names - certain reserved variable names that may not be declared but which are set in an initial state. They may, however, be accessed and modified.

The complete language definition looks as follows

```
type aExp =
| N of int           (* Integer literal *)
| V of string        (* Variable reference *)

| WL                 (* Word length *)
| PV of aExp         (* Point value lookup at word index *)

| Add of aExp * aExp (* Addition *)
| Sub of aExp * aExp (* Subtraction *)
| Mul of aExp * aExp (* Multiplication *)
| Div of aExp * aExp (* NEW: Division *)
| Mod of aExp * aExp (* NEW: Modulo *)

| CharToInt of cExp  (* Cast to integer *)

and cExp =
| C of char          (* Character literal *)
| CV of aExp          (* Character lookup at word index *)

| ToUpper of cExp     (* Convert character to upper case *)
| ToLower of cExp     (* Convert character to lower case *)

| IntToChar of aExp   (* Cast to character *)

type bExp =
| TT                 (* True *)
| FF                 (* False *)

| AEq of aExp * aExp (* Numeric equality *)
| ALt of aExp * aExp (* Numeric less than *)
```

```

| Not of bExp          (* Boolean not *)
| Conj of bExp * bExp (* Boolean conjunction *)

| IsLetter of cExp      (* Check for letter *)
| IsDigit of cExp       (* Check for digit *)

type stm =
| Declare of string      (* NEW: Variable declaration *)
| Ass of string * aExp    (* variable assignment *)
| Skip                (* Nop *)
| Seq of stm * stm       (* Sequential composition *)
| ITE of bExp * stm * stm (* If-Then-Else statement *)
| While of bExp * stm     (* While statement *)

```

These new statements and operators have the possibility of going wrong - a program can, for instance, divide by zero, declare a variable that has already been declared, or try to reference a variable that has not been declared. Exactly what we consider to be errors will be made clear in Section 1.

To write the evaluation functions for this program we will use railway-oriented programming. More precisely we will use a combination of the state and the error monad as demonstrated in the lecture.

1. Create the auxiliary functions needed to work effectively with changing state and errors.
2. Write evaluation functions for our DSL using these functions and railway-oriented programming
3. Write the same functions as in 2 but using computation expressions.
4. Write functions to help model the board using either the results from 2 or 3.

Note that 4 does not require 3 to do. It is there for you to practice on computational expressions. We recommend that you do 4 before attempting 3.

For this assignment you will be given an F# project that you can load. It contains the code for the state monad in a standalone module to ensure that your evaluation functions for parts 2-4 do not break abstraction. We have also included test cases that we present here for your functions. To run these tests just execute the project. They are also on CodeJudge.

If you have problems loading the project you can create your own and import the .fs- and the .fsi-files. Remember that order is important in F# projects. The order used is `StateMonad.fsi`, `StateMonad.fs`, `Eval.fs`, and `Program.fs`. This is also the order to use if you compile from the command line.

Before you start: Insert your definition of `hello` from Assignment 2.11 in `Eval.fs` at the marked place.

1: Handling errors and state

The answers to these assignments go in `StateMonad.fs`. Its interface, which you do not have to touch, is in `StateMonad.fsi`.

Before we start, we need to make clear what errors we support and what our states are. We will start with the errors.

```

type Error =
  | VarExists of string      (* Declaring an existing variable in the
                              same scope *)
  | VarNotFound of string    (* Referencing a variable that
                              does not exist *)
  | IndexOutOfBounds of int (* Trying to get the point or character value
                              that is outside the index of the word *)
  | DivisionByZero           (* Division or modulo by zero *)
  | ReservedName of string   (* Trying to declare a reserved name *)

```

The state is a record type that stores a stack of maps of variables, the word, and a set of reserved variable names. Note that this is different from our previous implementations where we stored the variable environment and the word separately. Our approach here keeps the state in one place.

```

type State = { vars      : Map<string, int> list
              word      : (char * int) list
              reserved   : Set<string> }

```

The intuition of `vars` is that it is a stack of variable bindings where an empty map (an empty variable environment) is pushed on the stack when entering a block (in a while-loop for instance) and popped when you exit the block, effectively forgetting any variable that was declared in the block. This will be made precise soon, but at a high level it works just like variable scoping in languages like Java or C# - variables declared within a loop or branch of an if-statement are:

- not available outside these sections
- shadow (replace) any already existing variable with the same name while inside these sections

The function `mkState : (string * int) list -> word -> string list -> State` takes an initial set of variable mappings `vs`, a word `w`, a list of reserved symbols `res` and returns a state with `vars` set to the singleton stack containing the names and values from `vs`, word set to `w`, and `reserved` set to `res` converted to a set.

```

let mkState vs w res =
  { vars = [Map.ofList vs];
    word = w;
    reserved = Set.ofList res }

```

As a consequence of the definition of `mkState` we know that the stack of variable environments `vars` will never be empty as it has a single environment when programs start running and a new empty environment is pushed when entering a while-loop or an if-statement, and popped when we exit the same. There must never be an occurrence of a pop that has not been preceded with a push. We will make this precise in Exercise 6.10.

For nearly all of the examples in this section we will be using the state

```

let state =
  mkState [("x", 5); ("y", 42)] hello (Set.ofList ["_pos_"; "_result_"])

```

where we use `hello` from Assignment 2.13. This is thus a state where the variable `x` is mapped to the value `5`, the variable `y` is mapped to the value `42`, the word being used is `hello`, and the reserved names are `_pos_` and `_result_` (similar to the states used in Assignment 3.7).

Finally, before we start with the actual assignments for this week, we need to set up the features we need for our railroad. This is very similar to the section on Parametric Results from the lecture.

```

type Result<'a, 'b> =
  | Success of 'a
  | Failure of 'b

type SM<'a> = S (State -> Result<'a * State, Error>)

let evalSM : (s : State) (a : SM<'a>) : Result<'a, Error> =
  match a s with
  | Success (result, _) -> Success result
  | Failure error -> Failure error

let bind (f : 'a -> SM<'b>) (a : SM<'a>) : SM<'b> =
  S (fun s ->
    match a s with
    | Success (av, s') ->
      let (S g) = f av
      g s'
    | Failure err -> Failure err)

let ret (v : 'a) : SM<'a> = S (fun s -> Success (v, s))
let fail err : SM<'a> = S (fun _ -> Failure err)

let (>=>) (a : SM<'a>) (f : 'a -> SM<'b>) : SM<'b> = bind f a
let (>>=>) (u : SM<unit>) (a : SM<'a>) : SM<'a> = u >=> (fun () -> a)

```

Note that the type for the state monad is encapsulated in an algebraic datatype with the constructor `S` rather than having a type alias. We do this to be able to put it into a `.fsi` file similar to what you have already done with your dictionaries and multisets for Assignment 4. Additionally, it cleans up the types considerably as `SM<'a>` is much shorter than `State -> Result<'a * State, Error>` which would have been the alternative.

Go back to the lecture slides for examples on how these connectives are used, but to briefly recapitulate:

- `(>=>)` has type `SM<'a> -> (a -> SM<'b>) -> SM<'b>` which given an expression `s a` and a function `f` evaluates `a` in some initial state `s` and
 - returns `Failure err` if `a s` evaluates to `Failure err`
 - returns `g s'` if `a s` evaluates to `Success av s'` and `f av` evaluates to `S g`
- `(>>=>)` has type `SM<unit> -> SM<'a> -> SM<'a>` which given an expression `s u` and an expression `s a` evaluates `u` in some initial state `s` and
 - returns `Failure err` if `u s` evaluates to `Failure err`

- returns `a s'` if `u s` evaluates to `Success () s'`

Both these connectives are used to compose expressions while hiding the otherwise manual book keeping required to keep the state up to date. A high-level view is that when they succeed

- `a >>= f` behaves similarly to `let x = a s in f x` for some initial state `s`
- `a >>>= b` behaves similarly to `a s; b` for some initial state `s`

In both these cases, evaluating `a s` updates the state which is passed on to `f` and `b` respectively. You can think of `>>>=` as a more advanced version of `;` in imperative languages like Java or C# - a programmable semicolon if you will. This will be made clear in the following examples.

Updating the state

We will now start creating infrastructure for updating and accessing the state of our programs. First, consider the following function `push : SM<unit>` that updates the variable state by pushing an empty map to the top of the stack. This function will be called whenever we enter a branch of an if-statement or a while loop.

This description may confuse you. After all, there is no state directly in our type. However, if you unfold the definition of `SM<'a>` you will find that any implementation of `push` must have the form `s f`, for some function `f`, where `f` has the type `state -> Result<unit * state, Error>`, and where the state is present both as an argument to the function `f` and in its return type for its `Success` case, allowing us to manipulate it. The reason we don't spell the types out this way is that one of the main selling points of railroad-oriented programming is that it allows us to hide the state within the types and to work at a higher level of abstraction.

```
let push : SM<unit> =
  S (fun s -> Success ((), {s with vars = Map.empty :: s.vars}))
//          ^^ This is your return value and the reason push
//          has type SM<unit>
```

Note that the return type is `SM<unit>` which means that there are no visible effects to this computation, but the stack is updated under the surface by modifying the initial state `s`.

The notation `{r with fieldName = x}` is used to create a copy of the record `r` except for the field `fieldName` which is set to `x` in stead of its original value. The rest of the record, no matter its size, is left unchanged.

A more complicated example is the function that looks up variables from the stack. The function `lookup : string -> SM<int>` takes a variable name `x` and looks up the first occurrence of the variable on the stack. If there is no such variable then the error `VarNotFound x` is returned.

```
let lookup (x : string) : SM<int> =
  let rec aux =
    function
    | [] -> None
    | m :: ms ->
      match Map.tryFind x m with
      | Some v -> Some v
```

```

      | None    -> aux ms

S (fun s ->
  match aux s.vars with
  | Some v -> Success (v, s)
    // The value v has type int so the return type is SM<int>.
    // The initial state s is not changed as variable
    // lookups do not change the state.
  | None    -> Failure (VarNotFound x))

```

The important thing to note here is again that the state does not appear in the type of the `lookup` function but is abstracted away from completely. We are just getting started, but even now we can create some simple tests and also demonstrate how to connect the two functions `push` and `lookup` that we have just created using `>>>=`.

```

> lookup "x" |>
  evalSM state;;
- val it : Result<int,Error> = Success 5

> lookup "z" |>
  evalSM state;;
- val it : Result<int,Error> = Failure (VarNotFound "z")

> push >>>= push >>>= lookup "y" |>
  evalSM state;;
- val it : Result<int,Error> = Success 42

> push >>>= push >>>= push >>>= lookup "z" |>
  evalSM state;;
- val it : Result<int,Error> = Failure (VarNotFound "z")

```

Note that our programs composed of `push` and `lookup` never mention the state. An initial state is only given when evaluating our program. Also note that since `push` has the type `SM<unit>` we use `>>>=` to compose it with the rest of the program and not `>>=`.

Green Exercises

Exercise 6.1

Create a function `pop : SM<unit>` that pops the top variable environment off the stack of the state. This function will be called when we exit a branch from an if-statement or a while loop. You do **not** have to consider the case when the stack is empty and can either have the program fail, throw an exception, or just not handle the case at all. Ending up in an empty stack is not the fault of the programmer, but an error in the evaluation functions that we write later on.

Examples:

```

> push >>>= pop >>>= lookup "x" |>
  evalSM state;;
- val it : Result<int,Error> = Success 5

> pop >>>= push >>>= lookup "x" |>
  evalSM state;;
- val it : Result<int,Error> = Failure (VarNotFound "x")

> pop >>>= pop >>>= lookup "x" |>
  evalSM state;;
- (* Fails in any way you like. If you simply ignore this case the
    program will fail automatically. This is perfectly fine. *)

```

Exercise 6.2

Create a function `wordLength : SM<int>` that returns the length of the word of the state and leaves the state unchanged.

Examples:

```

> wordLength |>
  evalSM state;;
- val it : Result<int,Error> = Success 5 // The length of HELLO is 5

```

Exercise 6.3

Create a function `characterValue : int -> SM<char>` that given a position `pos` returns the character at position `pos` of the word in the state. If `pos` is not a valid word index then fail with `IndexOutOfBounds pos`. The state itself is left unchanged.

Examples:

```

> characterValue 0 |> evalSM state;;
- val it : Result<char,Error> = Success 'H'

> characterValue 4 |> evalSM state;;
- val it : Result<char,Error> = Success 'O'

> lookup "y" >>= characterValue |> evalSM state;;
- val it : Result<char,Error> = Failure (IndexOutOfBounds 42)

```

Exercise 6.4

Create a function `pointValue : int -> SM<int>` that given a position `pos` returns the point value of the character at position `pos` of the word in the state. If `pos` is not a valid word index then return `IndexOutOfBounds pos`. The state itself is left unchanged.

Examples:

```

> pointValue 0 |> evalSM state;;
- val it : Result<char,Error> = Success 4

> pointValue 3 |> evalSM state;;
- val it : Result<char,Error> = Success 1

> lookup "y" >>= pointValue |> evalSM state;;
- val it : Result<char,Error> = Failure (IndexOutOfBounds 42)

```

Yellow Exercise

Exercise 6.5

Using the function `lookup` for inspiration, create a function `update : string -> int -> SM<unit>` that given a variable name `x` and a value `v` updates the first occurrence of `x` in the stack with the variable `v`. The rest of the state is left untouched. If `x` does not appear in the stack, the error `VarNotFound x` is returned. Since the stack can never be empty you do not have to take this case into consideration and you may do whatever you like. Failing is one option, but the easiest is most likely to return `VarNotFound x`, which is what `lookup` does.

```

> update "x" 7 >>>= lookup "x" |>
  evalSM state;;
- val it : Result<int,Error> = Success 7

> push >>>= update "x" 7 >>>= lookup "x" |>
  evalSM state;;
- val it : Result<int,Error> = Success 7

> push >>>= update "x" 7 >>>= pop >>>= lookup "x" |>
  evalSM state;;
- val it : Result<int,Error> = Success 7

> pop >>>= update "x" 7 >>>= push >>>= lookup "y" |>
  evalSM state;;
- val it : Result<int,Error> = Failure (VarNotFound "x")

> lookup "x" >>=
  (fun v1 -> lookup "y" >>=
    (fun v2 -> update "x" (v1 + v2))) >>>=
  lookup "x" |>
  evalSM state;;
- val it : Result<int,Error> = Success 47

> lookup "x" >>=
  (fun v1 -> lookup "y" >>=
    (fun v2 -> update "x" (v1 + v2))) >>>=
  lookup "y" |>

```



```
evalSM state;;  
- val it : Result<int,Error> = Success 42
```

Please remember the last two examples and see how we use `>=>`, rather than `>>>=` to use the result for `lookup` in later parts of a program.

Exercise 6.6

Create a function `declare : string -> SM<unit>` that given a variable name `x` adds that variable to the top variable environment of the stack with a starting value of `0`. In addition

1. If the variable has the same name as one of the reserved words, fail with `ReservedName x`.
2. If the variable already exists in the top variable environment, fail with `VarExists x`
3. If the stack is empty then the function is undefined and you can do anything you like, including failing or returning the *wrong* answer.

Examples:

```
> declare "z" >>>= lookup "z" |>  
evalSM state;;  
- val it : Result<int,Error> = Success 0  
  
> declare "z" >>>= update "z" 123 >>>= lookup "z" |>  
evalSM state;;  
- val it : Result<int,Error> = Success 123  
  
> declare "x" >>>= lookup "x" |>  
evalSM state;;  
- val it : Result<int,Error> = Failure (VarExists "x")  
  
> declare "z" >>>= declare "z" |>  
evalSM state;;  
- val it : Result<unit,Error> = Failure (VarExists "z")  
  
> declare "z" >>>= update "z" 123 >>>= push >>>=  
declare "z" >>>= update "z" 456 >>>= lookup "z" |>  
evalSM state;;  
- val it : Result<int,Error> = Success 456  
  
> declare "z" >>>= update "z" 123 >>>= push >>>=  
declare "z" >>>= update "z" 456 >>>= pop >>>=  
lookup "z" |>  
evalSM state;;  
val it : Result<int,Error> = Success 123  
  
> declare "_pos_" >>>= lookup "_pos_" |>  
evalSM state;;  
- val it : Result<int,Error> = Failure (ReservedName "_pos_")
```

2: Evaluation functions using railway-oriented programming

The answers to these assignments go in `Eval.fs`.

We will now create an evaluation function for our domain specific language using the constructs you have just created above. You may **not** place these functions in `StateMonad.fs` but in `Eval.fs` which imports `StateMonad.fsi`. We have now contained our plumbing in one place and we no longer have to concern ourselves with how the state is updated internally.

Green Exercises

Assignment 6.7

Create a function `add : SM<int> -> SM<int> -> SM<int>` that given two numbers `a` and `b` returns `a + b`. The state does not change. You only need to use `>>=`, `ret`, and standard arithmetic to construct your function.

Hint: Look at the last two examples of Exercise 6.5 and/or the lecture slides for inspiration.

Examples:

```
> add (ret 5) (ret 7) |>
  evalSM state;;
- val it : Result<int,Error> = Success 12

> add (lookup "x") (lookup "y") |>
  evalSM state;;
- val it : Result<int,Error> = Success 47

> add wordLength (lookup "z") |>
  evalSM state;;
- val it : Result<int,Error> = Failure (VarNotFound "z")
```

Assignment 6.8

Create a function `div : SM<int> -> SM<int> -> SM<int>` that given two numbers `a` and `b` returns the integer division `a / b` if `b` is not equal to `0` and fails with `DivisionByZero` otherwise. The state does not change. You only need to use `>>=`, `ret`, `fail`, possibly an if-statement, and standard arithmetic to construct your function.

Examples:

```
> div (ret 7) (ret 5) |>
  evalSM state;;
- val it : Result<int,Error> = Success 1

> div (lookup "y") (lookup "x") |>
```

```

evalSM state;;
- val it : Result<int,Error> = Success 8

> div wordLength (lookup "z") |>
evalSM state;;
- val it : Result<int,Error> = Failure (VarNotFound "z")

> declare "z" >>= div (lookup "x") (lookup "z") |>
evalSM state;;
- val it : Result<int,Error> = Failure DivisionByZero

```

Assignment 6.9

Create functions `arithEval : aExp -> SM<int>`, `charEval : cExp -> SM<char>` and `boolEval : bExp -> SM<bool>` that given an expression evaluates that expression in the input state. You may declare helper functions (as in Assignments 6.7 and 6.8) if you want, or inline them as every individual case is small. You may never expose the state but all state manipulation must be done via the functions described in Section 1. You will, however, find functions that you have created for previous versions of these evaluation functions (like `isLetter` or `isDigit`) from Assignment 3 useful.

The following errors should be raised:

1. `DivisionByZero` whenever we divide or use modulo by `0`.
2. `VarNotFound x` whenever we reference a variable `x` that does not exist in the state.
3. `IndexOutOfBounds x` whenever we try to get a point value or a character from the word at an invalid index `x`.

Note that errors 2 and 3 are all handled by the functions that we created in Section 1 and will be propagated automatically as long as you use `>>=` and `>>>=` properly.

Examples:

```

> arithEval (V "x" .+. N 10) |>
evalSM state;;
- val it : Result<int,Error> = Success 15

> arithEval (WL .*. N 10) |>
evalSM state;;
- val it : Result<int,Error> = Success 50

> arithEval (CharToInt (CV (N 0))) |>
evalSM state;;
- val it : Result<int,Error> = Success 72

> arithEval (PV (N -5)) |>
evalSM state;;
- val it : Result<int,Error> = Failure (IndexOutOfBounds -5)

> arithEval (V "x" .%. N 0) |>

```

```
evalSM state
- val it : Result<int,Error> = Failure DivisionByZero
```

```
> charEval (C 'H') |>
evalSM state;;
- val it : Result<char,Error> = Success 'H'

> charEval (ToLower (CV (N 0))) |>
evalSM state;;
- val it : Result<char,Error> = Success 'h'

> charEval (ToUpper (C 'h')) |>
evalSM state;;
- val it : Result<char,Error> = Success 'H'

> charEval (CV (V "x" .-. N 1)) |>
evalSM state;;
- val it : Result<char,Error> = Success 'O'
```

```
> boolEval TT |>
evalSM state;;
- val it : Result<bool,Error> = Success true

> boolEval FF |>
evalSM state;;
- val it : Result<bool,Error> = Success false

> boolEval ((V "x" .+. V "y") .=. (V "y" .+. V "x")) |>
evalSM state;;
- val it : Result<bool,Error> = Success true

> boolEval ((V "x" .+. V "y") .=. (V "y" .-. V "x")) |>
evalSM state;;
- val it : Result<bool,Error> = Success false

> boolEval (IsDigit (CV (V "x"))) |>
evalSM state;;
- val it : Result<bool,Error> = Failure (IndexOutOfBounds 5)

> boolEval (IsLetter (CV (V "x" .-. N 1))) |>
evalSM state;;
- val it : Result<bool,Error> = Success true

> boolEval (IsDigit (CV (N 0))) |>
evalSM (mkState [] [( '0', 42)] []);;
- val it : Result<bool,Error> = Success true
```

Yellow Exercises

Assignment 6.10

Create a function `evalStmnt : stm -> SM<unit>` that given a statement `stm` evaluates that statement. This function has the same specification as the one you wrote in 3.7 but with three additions:

1. `Declare x` declares a new variable `x` with a default value of `0` in the top variable environment on the stack. Moreover,
 - If `x` is equal to one of the reserved names then fail with `ReservedName x`
 - If `x` is already declared in the top environment then fail with `VarExists x`
2. Whenever you enter the branch of an if-statement or a while loop you must push an empty environment to the top of the stack (use `push` and `>>>=`).
3. Whenever you exit the branch of an if-statement or a while loop you must pop the top environment from the stack (use `pop` and `>>>=`).

Note that the errors for `Declare` are already handled by the `declare` function that you wrote in Assignment 6.3 and you do not have to do anything extra here.

For some of the examples we will use the following empty state.

```
let emptyState = mkState [] [] Set.empty
```

Examples:

```
> stmtEval (Ass ("x", N 5)) |>
  evalSM emptyState;;
- val it : Result<(), Error> = Failure (VarNotFound "x")

> stmtEval (Seq (Declare "x", Ass ("x", N 5))) >>>= lookup "x" |>
  evalSM emptyState;;
- val it : Result<int, Error> = Success 5

> stmtEval
  (Seq (Declare "x",
        Seq (Declare "y",
              Seq (Ass ("x", WL),
                    Ass ("y", N 7))))) >>>=
  lookup "x" >>= (fun vx -> lookup "y" >>= (fun vy -> ret (vx, vy))) |>
  evalSM emptyState;;
- val it : Result<(int * int), Error> = Success (0, 7)

> stmtEval
  (Seq (Declare "x",
        Seq (Declare "y",
              Seq (Ass ("x", WL),
                    Ass ("y", N 7))))) >>>=
  lookup "x" >>= (fun vx -> lookup "y" >>= (fun vy -> ret (vx, vy))) |>
```

```
evalSM state;; (* Running in state, where x is already declared, rather than
emptyState *)
- val it : Result<(int * int),Error> = Failure (VarExists "x")
```

3: Evaluation functions using computational expressions

Assignment 6.11

Create functions `arithEval2 : aExp -> SM<int>`, `charEval2 : cExp -> SM<char>`, `boolEval2 : bExp -> SM<bool>`, and `stmtEval2 : stmt -> SM<unit>` that work in the same way as their corresponding functions from 2 but that use computational expressions in stead.

To set things up you may use the following builder:

```
type StateBuilder() =

member this.Bind(x, f)      = x >>= f
member this.Return(x)      = ret x
member this.ReturnFrom(x)  = x
member this.Delay(f)       = f ()
member this.Combine(a, b)  = a >>= (fun _ -> b)

let prog = new StateBuilder()
```

Recall from the lecture that:

1. `let! x = a` corresponds to `>>=`. It evaluates an expression of type `SM<'a>` and stores the result of type `'a` in `x`.
2. `do! a` corresponds to `>>>=`. It evaluates an expression of type `SM<unit>` and continues with any updates to the state being passed along.
3. `return x` corresponds to `ret`. Given a return value of type `'a` it returns the corresponding value of type `SM<'a>`.
4. `return! a` simply returns `a` that has type `SM<'a>` directly.

For instance, the expression

```
declare "x" >>>= update "x" 5 >>>=
declare "y" >>>= lookup "x" >>=
(fun vx -> lookup "y" (fun vy -> ret (vx, vy)))
```

can be written as

```

prog {
  do! declare "x"
  do! update "x" 5
  do! declare "y"
  let! vx = lookup "x"
  let! vy = lookup "y"

  return (vx, vy)
}

```

and both expressions will evaluate to

```
Success (5, 0)
```

4: Modelling the Scrabble board

This assignment will be used for the Scrabble project. These assignments are all very small. Both 6.12 and 6.13 boil down to evaluating a statement using `stmtEval` in a specific initial state, that you can create using `mkState`, and then doing a lookup for the `_result_` variable and finally getting the result using `evalSM`. This is the same pattern that we have used in nearly every single example so far.

Red Exercises

Assignment 6.12

We will now remake assignment 3.7 in our new system.

Recall that we had the type `squareFun` where

```
type squareFun = word -> int -> int -> int
```

and the arguments are a word, a position, and an accumulator respectively.

For this assignment we will change it to allow for failures and set the type to

```
type squareFun = word -> int -> int -> Result<int, Error>
```

Create a function `stmtToSquareFun : stmt -> squareFun` that given a statement `stm` returns a function that given a word `w`, a position `pos`, and an accumulator `acc` evaluates `stm` in an initial state (created using `mkState`) where:

- The variable environment contains the variables `_pos_`, `_acc_`, and `_result_` where
 - the variable `_pos_` is initialised to `pos`
 - the variable `_acc_` is initialised to `acc`
 - the variable `_result_` is initialised to `0`
- the word is set to `w`

- the restricted names are `_pos_`, `_acc_`, and `_result_`.

The function should return the value of the `_result_` variable of the final state when the statement has been executed and fail if the evaluation fails. You should use the evaluation function from either Section 2 or Section 3.

Using your function we can create similar square functions as in 3.7.

```
let arithSingleLetterScore = PV (V "_pos_") .+. (V "_acc_")
let arithDoubleLetterScore = ((N 2) .*. PV (V "_pos_")) .+. (V "_acc_")
let arithTripleLetterScore = ((N 3) .*. PV (V "_pos_")) .+. (V "_acc_")

let arithDoubleWordScore = N 2 .*. V "_acc_"
let arithTripleWordScore = N 3 .*. V "_acc_"

let stmtSingleLetterScore = Ass ("_result_", arithSingleLetterScore)
let stmtDoubleLetterScore = Ass ("_result_", arithDoubleLetterScore)
let stmtTripleLetterScore = Ass ("_result_", arithTripleLetterScore)
let stmtDoubleWordScore = Ass ("_result_", arithDoubleWordScore)
let stmtTripleWordScore = Ass ("_result_", arithTripleWordScore)

let singleLetterScore = stmtToSquareFun stmtSingleLetterScore
let doubleLetterScore = stmtToSquareFun stmtDoubleLetterScore
let tripleLetterScore = stmtToSquareFun stmtTripleLetterScore
let doubleWordScore = stmtToSquareFun stmtDoubleWordScore
let tripleWordScore = stmtToSquareFun stmtTripleWordScore

let containsNumbers =
  stmtToSquareFun
    (Seq (Declare "i",
      (Seq (Ass ("_result_", V "_acc_"),
        While (V "i" .<. WL,
          ITE (IsDigit (CV (V "i")),
            Seq (
              Ass ("_result_", V "_result_" .*. N -1),
              Ass ("i", WL)),
            Ass ("i", V "i" .+. N 1))))))))
```

Examples:

```
> singleLetterScore hello 0 0;;
- val it : int = Success 4

> doubleLetterScore hello 0 0;;
- val it : int = Success 8

> tripleLetterScore hello 0 0;;
- val it : int = Success 12
```



```

> singleLetterScore hello 0 42;;
- val it : int = Success 46

> doubleLetterScore hello 0 42;;
- val it : int = Success 50

> tripleLetterScore hello 0 42;;
- val it : int = Success 54

> containsNumbers hello 5 50;;
- val it : int = Success 50

> containsNumbers (('0', 100)::hello) 5 50;;
- val it : int = Success -50

> containsNumbers (hello @ [('0', 100)]) 5 50;;
- val it : int = Success -50

```

Assignment 6.13

So far we have only modelled squares on the board. We will now model the entire board. At their core, boards are modelled as functions from coordinates to square function options.

```

type coord = int * int

type boardFun = coord -> Result<squareFun option, Error>

```

The intuition behind this function is that we have a coordinate `coord` that represents the `x` and the `y` coordinate of a board and that the board function returns the square at a certain coordinate and `None` if the square is blank. For a standard board any coordinate outside the board would be empty, but we do support infinite boards as well, or boards with big holes in them.

To create a board functions we will again use our DSL above in a similar way to how we create square functions. This time around, however, the function takes the arguments `_x` and `_y` to represent the coordinate and a return variable `_result` that is an integer representing which square is placed at that coordinate. We will use a map of type `Map<int, squareFun>` to obtain a concrete square from the integer stored in `_result`.

The standard Scrabble board is defined in the following way:

```

let abs v result = ITE (v <. N 0, Ass (result, v *. N -1), Ass (result, v))

let twsCheck x y = ((V x ==. N 0) .&&. (V y ==. N 7)) .||.
  ((V x ==. N 7) .&&. ((V y ==. N 7) .||. (V y ==. N 0)))

let dwsCheck x y = (V x ==. V y) .&&. (V x <. N 7) .&&. (V x >. N 2)

let tlsCheck x y = ((V x ==. N 6) .&&. (V y ==. N 2)) .||.

```

```

      ((V x .=. N 2) .&&. ((V y .=. N 2) .||. (V y .=. N 6)))

let dlsCheck x y = ((V x .=. N 0) .&&. (V y .=. N 4)) .||.
  ((V x .=. N 1) .&&. ((V y .=. N 1) .||. (V y .=. N 5))) .||.
  ((V x .=. N 4) .&&. ((V y .=. N 0) .||. (V y .=. N 7))) .||.
  ((V x .=. N 5) .&&. (V y .=. N 1)) .||.
  ((V x .=. N 7) .&&. (V y .=. N 4))

let insideCheck x y = ((V x .<. N 8) .&&. (V y .<. N 8))

let checkSquare f v els = ITE (f "xabs" "yabs", Ass ("_result_", N v), els)

let standardBoard =
  Seq (Declare "xabs",
    Seq (Declare "yabs",
      Seq (abs (V "_x_") "xabs",
        Seq (abs (V "_y_") "yabs",
          checkSquare twsCheck 4
            (checkSquare dwsCheck 3
              (checkSquare tlsCheck 2
                (checkSquare dlsCheck 1
                  (checkSquare insideCheck 0
                    (Ass ("_result_", N -1))))))))))

let boardMap = [(0, singleLetterScore); (1, doubleLetterScore); (2, tripleLetterScore);
  (3, doubleWordScore); (4, tripleWordScore)] |> Map.ofList

```

Here, the statement `standardBoard` stores a number between 0 and 4 in `_result_` and provides the mapping to the corresponding squares in `boardMap`.

Create a function `stmtntToBoardFun : stmtnt -> Map<int, squareFun> -> boardFun` that given a statement `stm` and a lookup table of identifiers and square functions `squares` runs `stm` in the state where:

- The variable environment contains the variables `_x_`, `_y_`, and `_result_` where:
 - the variables `_x_` and `_y_` are initialised to be the x- and the y-values of the coordinate to the board function
 - the variable `_result_` is initialised to 0
- The word is set to the empty word (we guarantee it will never be used when evaluating boards)
- `_x_`, `_y_`, and `_result_` are reserved words.

after which it looks up the integer value `id` stored in `_result_` and returns

- `Success (Some sf)`, if looking up the key `id` in `squares` results in `sf`
- `Success None` if the key `id` does not exist in `squares`
- Fails if the evaluation of `stm` fails.

It is important to note that a result of `Success None` is not considered a failure, it just means that the coordinate is empty (outside the board, for instance).

```
Program.fs )
```

```
printfn " "
```

and get the result

[illegible]

where empty squares are marked with #.

Assignment 6.14

Boards are modelled using the following record type:

```
type board {  
  center      : coord  
  defaultSquare : squareFun  
  squares     : boardFun  
}
```

Every board has a center coordinate `center` over which the first word must be placed. Moreover it has a default square, which is the square that is used whenever a letter has already been placed on the board. In standard scrabble this would be the Single Letter Square, as any tile placed on that square can be counted as a single letter for words built later no matter what the original square type was. Finally, the board itself is modelled as a `boardFun` described above.

Create a `function mkBoard : coord -> stmtnt -> stmtnt -> (int * stmtnt) list -> board` that given

- a center coordinate `c`
- a statement representing the default square `defaultSq`
- a statement `boardStmtnt` representing the board
- a list of identifiers and statements `ids` representing the different squares on the board

returns the board that has

```
* the center coordinate `c`  
* the default square `defaultSq` compiled with `stmtntToSquareFun`
```

- the squares `boardStmtnt` compiled with `stmtntToBoardFun` where the square lookup map is derived from `ids` by
 - mapping all pairs `(k, sq)` in `ids` to `(k, sq')` where `sq'` is `sq` compiled with `stmtntToSquareFun` (use `List.map`)
 - converting the resulting list to a map (use `Map.ofList`)

The following code creates the standard board

```
let ids =  
  [(0, stmtntSingleLetterScore); (1, stmtntDoubleLetterScore); (2,  
  stmtntTripleLetterScore);  
  (3, stmtntDoubleWordScore); (4, stmtntTripleWordScore)]  
  
let standardBoard =  
  mkBoard (0, 0) stmtntSingleLetterScore standardBoardFun ids
```

and the following test code outputs the same result as above.

```
for y in -10..10 do
  for x in -10..10 do
    printf "%s " (standardBoard.squares (x, y) |> evalSquare hello 1 3 |> toString)
  printfn ""
```